A Manual for BASIC, the elementary algebraic language designed for use with the Dartmouth Time Sharing System.

1 October 1964

Copyright 1964 by the Trustees of Dartmouth College. Reproduction for non-commercial use is permitted provided due credit is given to Dartmouth College.
The development of the BASIC Language, and of this Manual, has been supported in part by the National Science Foundation under the terms of Grant NSF GE 3864.
# TABLE OF CONTENTS

I  What is a Program  

II  A BASIC Primer  

2.1  An Example  

2.2  Expressions, Numbers, and Variables  

2.3  Loops  

2.4  Use of the Time Sharing System  

2.5  Errors and Debugging  

2.6  Summary of Elementary BASIC Statements  

III  More Advanced BASIC  

3.1  More About PRINT  

3.2  Lists and Tables  

3.3  Functions and Subroutines  

3.4  Some Ideas for More Advanced Programmers  

IV  Card BASIC  

4.1  Purpose  

4.2  How to Prepare a Deck  

4.3  Differences in Operation  

4.4  MAT  

V  Appendices  

A.  Error Stops  

B.  Limitations on BASIC  

C.  The 15 BASIC Statements  

I

WHAT IS A PROGRAM?

A program is a set of directions, a recipe, that is used to provide an answer to some problem. It usually consists of a set of instructions to be performed or carried out in a certain order. It starts with the given data and parameters as the ingredients, and ends up with a set of answers as the cake. And, as with ordinary cakes, if you make a mistake in your program, you will end up with something else -- perhaps hash!

Any program must fulfill two requirements before it can even be carried out. The first is that it must be presented in a language that is understood by the "computer." If the program is a set of instructions for solving linear equations, and the "computer" is a person, the program will be presented in some combination of mathematical notation and English. If the person solving the equations is a Frenchman, the program must be in French. If the "computer" is a high speed digital computer, the program must be presented in a language the computer can understand.

The second requirement for all programs is that they must be completely and precisely stated. This requirement is crucial when dealing with a digital computer, which has no ability to infer what you meant -- it can act only upon what you actually present to it.

We are of course talking about programs that provide numerical answers to numerical problems. To present a program in the English language, while easy on the programmer, poses great difficulties for the computer because English, or any other spoken language, is rich with
ambiguities and redundancies, those qualities which make poetry possible but computing impossible. Instead, you present your program in a language that resembles ordinary mathematical notation, which has a simple vocabulary and grammar, and which permits a complete and precise specification of your program. The language that you will use is BASIC (Beginner's All-purpose Symbolic Instruction Code) which is at the same time precise, simple, and easy to understand.

Your first introduction to the BASIC language will be through an example. Next you will learn how to use the Dartmouth Time Sharing System to execute BASIC programs. Finally, you will study the language in more detail with emphasis on its rules of grammar and on examples that show the application of computing to a wide variety of problems.
2.1  **An Example**

The following example is a complete BASIC program for solving two simultaneous linear equations in two unknowns with possibly several different right hand sides. The equations to be solved are

\[ A_1 x_1 + A_2 x_2 = B_1 \]
\[ A_3 x_1 + A_4 x_2 = B_2 \]

Since there are only two equations, we may find the solution by the formulas

\[ x_1 = \frac{(B_1 A_4 - B_2 A_2)}{(A_1 A_4 - A_3 A_2)} \]
\[ x_2 = \frac{(A_1 B_2 - A_3 B_1)}{(A_1 A_4 - A_3 A_2)} \]

It is noted that a unique solution does not exist when the denominator \( A_1 A_4 - A_3 A_2 \) is equal to zero. Study the example carefully -- in most cases the purpose of each line in the program is self-evident.

10 READ A1, A2, A3, A4
15 LET D = A1 * A4 - A3 * A2
20 IF D = 0 THEN 65
30 READ B1, B2
37 LET X1 = (B1*A4 - B2 * A2 ) / D
42 LET X2 = ( A1 * B2 - A3 * B1)/D
55 PRINT X1, X2
60 GOTO 30
65 PRINT "NO UNIQUE SOLUTION"
70 DATA 1, 2, 4
80 DATA 2, -7, 5
85 DATA 1, 3, 4, -7
90 END
We immediately observe several things about the above sample program. First, all lines in the program start with a line number. These serve to identify the lines in the program, each one of which is called a statement; thus, a program is made up of statements, most of which are instructions to be performed by the computer. These line numbers also serve to specify the order in which the statements are to be performed by the computer, which means that you could type your program in any order. Before the program is run by the computer, it sorts out and edits the program, putting the statements into the order specified by their line numbers. (This editing process makes extremely simple the correcting and changing of programs, as will be explained in later sections.)

The second observation is that each statement starts, after its line number, with an English word. This word denotes the type of the statement. There are fifteen types of statements in BASIC, nine of which are discussed in this chapter. Of these nine, seven appear in the sample program above.

The third observation is that we use only capital letters, and that the letter "Oh" is distinguished from the numeral "Zero" by having a diagonal slash through the "Oh". This feature is made necessary by the fact that in a computer program it is not always possible to tell from the context whether the letter or the number was intended unless they have a different appearance. This distinction is made automatically by the teletype machine, which also has a special key for the number "One" to distinguish it from the letter "Eye" or lower case "L".

A fourth observation, though perhaps less obvious than the first three, is that spaces have no significance in BASIC (except in messages
to be printed out as shown in statement number 65.) Thus, spaces may be used, or not used, at will to "pretty up" a program and make it more readable. For instance, statement 15 could have been typed as

\[ \text{15 LETD} = A1 \times A4 - A2 \times A3 \]

a fully equivalent though less readable form.

Turning now to the individual statements in the program, we observe that the first statement, numbered 10, is a READ statement. When the computer encounters a READ statement while executing your program, it will cause the variables whose names are listed after the READ to be given values according to the next available numbers in the DATA statements. Thus, in the example, when statement 10 is first encountered, it will cause the variable \( A1 \) to be given the value 1, the variable \( A2 \) to be given the value 2, the variable \( A3 \) to be given the value 4, and the variable \( A4 \) to be given the value 2.

The next statement, numbered 15, is a LET statement. It causes the computer to compute the value of the expression \( A1 A4 - A3 A2 \), and to assign this value to the variable \( D \). The expression computed in a LET statement can range from the very simple (consisting of only a single variable) to the very complex. The rules for forming these expressions are given in detail in the next section, but for now we point out that:

1. Variable names consist of a single capital letter possibly followed by a single digit;
2. The symbol * (asterisk) is always used to denote multiplication;
3. Parentheses may be needed to specify the order of the computation because the entire expression must appear on a single line;
4. No subscripts or superscripts as such are permitted, also because the expression must appear on a single line.

In line 20 the computer asks a question: "Is D equal to 0?" If the answer is yes, then the next statement to be executed by the computer is the one numbered 65. If the answer is no, the computer continues to statement 30, the next higher numbered one after 20.

In line 30 the computer causes the variables B1 and B2 to be given the values next appearing in the DATA statements elsewhere in the program. Since the first four data have already been used up, B1 is given the fifth value -7, and B2 is given the sixth value 5.

The statements numbered 37 and 42 complete the computation of the solution, X1 and X2. Notice that the denominator has been previously evaluated as the variable D. Thus it is not necessary to repeat the formula given in statement 15. Notice also how parentheses are used to specify that the numerator of the fraction consists of the entire quantity B1*A4 - B2*A2. If the parentheses had been omitted by mistake, the expression computed would have been B1*A4 - \( \frac{B2*A2}{D} \), which is incorrect.

Now that the answers have been computed, they will be printed out for you to see when the computer encounters statement 55. Notice that the comma is used to separate the individual items in the list of quantities to be printed out at that time.

Having completed the computation, statement 60 tells the computer to execute next statement number 30. We observe that the second encounter of statement 30 will cause the variables B1 and B2 to be given the values 1 and 3, respectively, the next available ones in the DATA statements.
After completing the computation for the second set of right hand sides and printing the answers, the computer will give the last values, 4 and -7, to the variable B1 and B2, compute and print the third set of answers, and then stop, because there is no more data when the READ statement 30 is encountered for the fourth time.

If D, the determinant of the coefficients, is zero, we know that the set of equations does not have a unique solution. In this case, statement 20 will cause the computer to execute statement 65 next. Statement 65 is again a PRINT statement, but instead of numerical answers being printed out, it will produce the English message

NO UNIQUE SOLUTION

We could have used any other recognizable message between the two quotation marks that would have indicated to us that no unique solution was possible for the given coefficients.

After printing the warning message the computer will execute next statement 90, an END statement, which stops the running of the program. (The running will also be stopped when a READ statement is encountered for which there is not sufficient data.) It is extremely important to remember that all programs must have an END statement, and that it must always be the highest numbered statement in the program. The intervening DATA statements are never executed by the computer; therefore, they may be placed anywhere in your program. The only requirement is that END be the highest numbered statement, including DATA statements, and that DATA statements are numbered in the order in which you wish the data to be used by the various READ statements in your program.
2.2 **Expressions, Numbers, and Variables**

Expressions in BASIC look like mathematical formulas, and are formed from numbers, variables, operations, and functions.

A number may contain up to nine digits with or without a decimal point, and possibly with a minus sign. For example, the following numbers are acceptable in BASIC:

5 2.5 123456789 .123456789 -123456

To extend the range of numbers, a factor of a power of ten may be attached, using the letter E to stand for "times ten to the power". Again, the following examples are all acceptable forms for the same number in BASIC,

-12.345 -12345 E-3 -.12345E2 -12345000E-6 -.00012345E5

It should be noted, however, that the E notation cannot stand alone; 1000 may be written 10E2 or 1E3 but not E3 (which looks like a variable and is so interpreted in BASIC.) It should also be noted that .000123456789 is illegal, and must be written as, say, .123456789E-3.

A variable in BASIC is denoted by any letter, or by any letter followed by a digit. For instance, these are acceptable variable names:

A X N5 X0 K9 $1

The difference between 0 and $, and between I and 1 should be observed. Thus, 10 is acceptable while any of 1$, 1$, and 10 are not (the last one is the number ten.)

A variable in BASIC stands for a number, usually one that is not
known to the programmer at the time the program was written. Variables are given or assigned values by LET and READ statements. The value so assigned will not change until the next time a LET or READ statement is encountered that names that variable.

Expressions are formed by combining variables and numbers together with arithmetic operations and parentheses just as in ordinary mathematical formulas. The symbols

\[ + \quad - \quad * \quad / \quad \uparrow \]

stand for "plus", "minus", "times", "divided by", and "to the power", respectively. Parentheses are used in the usual way, as in

\[(A1 + X)*(B - C\uparrow D7)\]

Because expressions must be presented as a single line, parentheses are often required where they might not be needed in ordinary mathematical notation. Thus,

\[ \frac{A - B}{C} \]

is written as \((A - B)/C\)

to show that the entire quantity \(A - B\) is to be divided by \(C\). Omitting the parentheses would leave \(A - B/C\), which is interpreted as \(A - (B/C)\).

Another example that arises quite often is

\[ \frac{A}{B*\uparrow C} \]

which is written as

\[ \begin{cases} 
\frac{A}{(B*\uparrow C)} \\
\frac{A}{B/\uparrow C} 
\end{cases} \]

\(A/B*\uparrow C\) will be interpreted the same as \((A/B)*\uparrow C\) or \((A*\uparrow C)/B\).

\(\uparrow\) The operation actually works with the absolute value of the left argument. Thus \(X\uparrow Y\) is interpreted as \(|X|\uparrow Y\). If \(X\) could be negative and you want \(X\uparrow 3\), you should write \(X*\uparrow X*\uparrow X\) or \(X*\uparrow X\uparrow 2\).
The way that expressions are interpreted can be summarized in terms of several rules, which correspond to standard mathematical notation. These are:

1. The expression inside a parentheses pair is computed before the parenthesized quantity is used in further computations.

2. Raising to a power is computed before multiply and/or divide, which in turn are computed before addition and/or subtraction, in the absence of parentheses.

3. Several multiply-divides, or several addition-subtractions, are computed from left to right.

The first rule tells us that in \((A + B) \times C\) we compute \(A + B\) first, then multiply the result by \(C\), an obvious interpretation. The second rule tells us that in \(A + B \times C \div D\) we first compute \(C \div D\), then multiply by \(B\), and finally add to \(A\). An equivalent expression is \(A + (B \times (C \div D))\).

The third rule states that \(A - B - C\) is interpreted as \((A - B) - C\) and not as \(A - (B - C)\). Applied to multiplies and divides, the rule tells us to interpret \(A \div B \div C\) as \((A \div B) \div C\) and not as \(A \div (B \div C)\). For raising to a power, \(A \uparrow B \uparrow C\) means \((A \uparrow B) \uparrow C\) or, equivalently, \(A \uparrow (B \times C)\). If you intend \(A \uparrow (B \uparrow C)\), you must use that form.

In addition to the arithmetic operations, some of the more common standard functions are available. For example, to compute \(\sqrt{1 + X^2}\) you would use \(\text{SQR}(1 + X \uparrow 2)\). The other standard functions are used in this same way, that is, the BASIC name of the function followed by the argument enclosed in parentheses.
<table>
<thead>
<tr>
<th>Function name</th>
<th>Purpose</th>
</tr>
</thead>
<tbody>
<tr>
<td>SIN(X)</td>
<td>sine of X</td>
</tr>
<tr>
<td>CØS(X)</td>
<td>cosine of X</td>
</tr>
<tr>
<td>TAN(X)</td>
<td>tangent of X</td>
</tr>
<tr>
<td>ATN(X)</td>
<td>arctangent (in radians) of X</td>
</tr>
<tr>
<td>EXP(X)</td>
<td>natural exponential of X, $e^X$</td>
</tr>
<tr>
<td>ABS(X)</td>
<td>absolute value of X, $</td>
</tr>
<tr>
<td>LØG(X)</td>
<td>natural logarithm of $</td>
</tr>
<tr>
<td>SQR(X)</td>
<td>square root of $</td>
</tr>
</tbody>
</table>

(Two other functions, RND(X) and INT(X), are explained in section 3.3)

The argument of a function may be any expression, no matter how complicated. For example

$$SQR( B \uparrow 2 - 4*A*C ) - 17$$
$$Z - EXP( X1 + LØG( A/X1 ) ) * TAN(A)$$
$$SQR( SIN(Q) \uparrow 2 + CØS(Q) \uparrow 2 )$$

are all acceptable in BASIC.

The use of the LØG and SQR functions requires a word of caution. In each case the argument is made positive before applying the function, since neither function is defined for negative arguments. Many times, though not always, an attempt to have the computer extract the square root of a negative number implies a fundamental error in the program. The user should be forewarned that such occasions, hopefully rare, may therefore be unnoticed.

The user may define new functions using the DEF statement, which is discussed in section 3.3.
2.3 Loops

Perhaps the single most important programming idea is that of a loop. While we can write useful programs in which each statement is performed only once, such a restriction places a substantial limitation on the power of the computer. Therefore, we prepare programs that have portions which are performed not once but many times, perhaps with slight changes each time. This "looping back" is present in the first program, which can be used to solve not one but many sets of simultaneous linear equations having the same left hand sides.

Making tables of, say, square roots is another example where a loop is necessary. Suppose that we wish to have the computer print a table of the first hundred whole numbers and their square roots. Without loops, one can easily see that a program would require 101 lines, all but the last having the form:

17 PRINT 17, SQR(17)

And if one wished to go not to 100 but to 50 only, a new program would be required. Finally, if one wanted to go to 10,000 the program would be absurd even if someone could be found to write it all down.

We notice that the basic computation, in this case a very simple printing, is practically the same in all cases -- only the number to be printed changes. The following program makes use of a loop.

10 LET X = 0
20 LET X = X + 1
30 PRINT X, SQR(X)
40 IF X <= 100 THEN 20
50 END

Statement 10, which gives to X the value 0, is the initialization of the loop. Statement 20, which increases the value of X by unity, is the statement that
insures that the loop is not merely repeating exactly the same thing -- an
infinite loop! Statement 30 is the body of the loop, the computation in which
we are interested. And statement 40 provides an exit from the loop after
the desired computation has been completed. All loops contain these four
characteristics: initialization, modification each time through the loop, the
body of the loop, and a way to get out.

Because loops are so important, and because loops of the type shown
in the example arise so often, BASIC provides two statements to enable one
to specify such a loop much more concisely. They are the FOR and the
NEXT statements, and would be used as follows in the example above:

10 FOR X = 1 TO 100
20 PRINT X, SQR(X)
30 NEXT X
40 END

Statement 10 contains both the initial and final values of X. Statement 30
specifies that X be increased to its next value. In this case, the value by
which X is increased each time is implied to be unity. If instead we wished
to print the square roots of the first 50 even numbers, we would have used

10 FOR X = 2 TO 100 STEP 2
20 PRINT X, SQR(X)
30 NEXT X
40 END

Omitting the STEP part is the same as assuming the step-size to be unity.

To print the square roots of the multiples of 7 that are less than
100, one might use for line number 10

10 FOR X = 7 TO 100 STEP 7

The loop will be performed for all values of X that are less than or equal
to 100, in this case, for X equal to 7, 14, ..., 91, 98.
2.4 Use of the Time Sharing System

The Dartmouth Time Sharing System consists of a large central computer with a number of input-output stations (currently, model 35 teletype machines.) Individuals using the input-output stations are able to "share" the use of the computer with each other in such a way as to suggest that they each have sole use of the computer. The teletype machines are the devices through which the user communicates with the computer.

Teletype machines are like ordinary typewriters, with certain modifications to make them suitable for transmitting messages over telephone lines. They have a nearly standard keyboard for letters and numbers, the most notable differences being that all letters are capitals and that the numeral one is not the same as the letter L. In addition there are several special characters which can be typed using either of the two "SHIFT" keys; these include the following special symbols that are used in BASIC programs:

```
+   -   *   /   ↑   =

(   )   <   >   .   ,   ;
```

There is a "CTRL" key that is related to standard teletype communications, but all the control symbols are ignored by BASIC. A layout of the keyboard is shown on the following page. It should be studied until the locations of these symbols are familiar.

There are three special keys that the user must know about.

"RETURN", which is located at the right hand end of the third row of keys, is the ordinary carriage return. More importantly, the computer ignores all typed lines until this key is pressed. It must be used after each line in a BASIC program, and after each line which
MODELS 33 and 35 KEYBOARD ARRANGEMENT

1 2 3 4 5 6 7 8 9 0 : =

ALT MODE Q W E R T Y U I O P

CTRL A S D F G H J K L

SHIFT Z X C V B N M , . / SHIFT

LOC LF
is a communication to the system. 

which is located on the letter "Oh" key while the "SHIFT" key is depressed, erases the last character typed. If the user notices that he has just mistyped a letter or a symbol, he pushes this key, which tells the computer to ignore the previously typed character. Pushing the backwards arrow more than once will delete the same number of characters, but only to the start of the line. For example, the sequence

ABCWTR←DE will appear as ABCDE

while

100 LET ← ← ← ← ← ← 200 LET X = Y

will appear as

200 LET X = Y

"ALT MODE", located at the left hand end of the third row of keys, is pressed to delete an entire typed line. It may be used at any time before a "RETURN" is used.

Besides the keyboard itself, there are four buttons that are needed to operate the teletype machine.

<table>
<thead>
<tr>
<th>BUTTON</th>
<th>FUNCTION</th>
</tr>
</thead>
<tbody>
<tr>
<td>&quot;ORIG&quot;</td>
<td>the leftmost of six small buttons on the right.</td>
</tr>
<tr>
<td>&quot;CLR&quot;</td>
<td>next to &quot;ORIG.&quot;</td>
</tr>
<tr>
<td>&quot;LOC LF&quot;</td>
<td>turns off the teletype.</td>
</tr>
<tr>
<td>&quot;BUZ-RLS&quot;</td>
<td>to the left of the space bar.</td>
</tr>
<tr>
<td></td>
<td>feeds the paper to permit tearing off.</td>
</tr>
<tr>
<td></td>
<td>turns off the buzzer, which goes on when the paper supply is low.</td>
</tr>
</tbody>
</table>

All other buttons and gadgets, including the telephone dial, are not connected.
When you sit down at a teletype machine, you must start by typing

HELLO

followed, as always, by a "RETURN". This starts the so-called HELLO sequence, a short series of questions and answers that serve to tell the computer who you are and what you wish to do.

Each user must have a user number. For students it is the six digit student ID number. The user number for any other person consists of a certain letter followed by five digits, and is assigned to the user by the Computation Center. When the computer asks for your user number, type it. (Don't forget the "RETURN").

The next information you must supply is the name of the system or language; in this case it will be BASIC.

When the computer asks about NEW or OLD, you type NEW unless you wish to retrieve an OLD program in order to continue working on it.

Finally, the computer asks for the name of your problem. You type any six letters, digits, or characters that you wish. If you are retrieving an OLD program, you must be careful to type the problem name exactly as you typed it originally.

The computer types READY, and is now ready for you to type in the new program, to add or change statements in an old program, or to command the computer to do something with your program. You must be careful to begin each statement in your program with a line number. These line numbers should:

-- contain no more than five digits,
After completely typing the program, you type

RUN

The computer will now analyze and run your program. It will then print a line that contains your user number, the problem name, the date, and the time of day. If there are no errors of form in the program, next will be printed the answers according to the PRINT statements within the program. (If the program runs for a long time, the teletype machine will periodically make little grinding noises, which indicate that no output has yet been produced. On the other hand, if a program is fairly short but produces lots of output, you may notice that the printing may break off after a while and commence somewhat later. During that interval, the computer is doing the computing necessary to produce the next batch of output.) Finally, a time statement is printed showing the total computing time used by the run. (In many cases it will show 0 seconds, indicating that the entire run required less than 0.5 seconds.)

A complete history of a successful run of the linear equation solver presented earlier is shown on the following page.

Besides HELLO and RUN, there are several extremely useful commands that may be given to the computer by typing at the start of a new line the command followed by pressing the "RETURN" key.

STOP causes the computer to stop whatever it is doing with the program. STOP may be used even with the computer is typing out; in this case it responds to your simply typing the letter S even without a "RETURN".

LIST will type out a complete listing of the program as it is, including all the corrections that have been made. To stop
HELLO
USER NUMBER--999999
SYSTEM--BASIC
NEW OR OLD--NEW
NEW PROBLEM NAME--LINEAR
READY.

10 READ A1, A2, A3, A4
15 LET D = A1 * A4 - A3 * A2
20 IF D = 0 THEN 65
30 READ B1, B2
35 LET X1 = (B1*A4 - B2 * A2) / D
40 LET X2 = ( A1 * B2 - A3 * B1)/D
55 PRINT X1, X2
60 G0 T0 30
65 PRINT "NO UNIQUE SOLUTION"
70 DATA 1, 2, 4
80 DATA 2, -7, 5
85 DATA 1, 3, 4, -7
90 END
RUN

USER NO. 999999  PROBLEM NAME: LINEAR  6 SEPT. 1964  TIME: 22:33

4      -5.5
0.666667  0.166667
-3.66667  3.83333

TIME:  1 SECS.
listing after it has started but before it is completed, type S.

LIST--XXXX will type out a listing of the program starting at line number XXXXX and continuing to the end or until the S key is pressed. For instance, LIST--70 in the linear equations problem will start listing at line 70, permitting the user to inspect the DATA statements without waiting for the early part of the program to be listed.

SAVE If the user is done working with a program at the moment and wishes to return to it at a later time, he should save it. For instance, if the user must leave the teletype and is only half finished with the original typing, he types SAVE. Later on, he retrieves exactly what he saved by typing OLD. He may then continue with the typing as if nothing had happened.

UNSAVE If a user has finished with a program that he has saved at some earlier time, he types UNSAVE. This action destroys the saved program, making room for other saved programs. All users are urgently requested to UNSAVE all programs for which they no longer have need.

CATALOG In case a user is working with several different programs and forgets what names he gave them, typing CATALOG requests the computer to list the names of all programs currently being saved by that user.

NEW Typing NEW will permit the programmer to start a new problem. This command may be given at any time, and has the effect of erasing the previous program (unless it was saved.) The computer will ask for the name of the new problem.

OLD This command is similar to NEW, but retrieves the named saved program, which may then be added to. Either OLD or NEW must be used in connection with the HELLO sequence, but either may be also used at any later time as well. Repeated use of OLD does not affect the saved program; it remains saved until it is unsaved, or until a new version is saved in place of it.

SCRATCH This command is very close to NEW in that it erases the previous work and presents a clean slate. It differs by retaining the name of the previous problem instead of asking for a new name.

RENAME serves to supply a new problem name for the current work without erasing it. It is useful if one wishes to save two similar versions of a program. Save the first, use RENAME, make the desired modifications, and then save or run the modified version.
When the user is finished with a session at the teletype, it is necessary only to leave the machine. The user should plan his session at the teletype to avoid long trances. If such happens, the user should save his work, and leave the machine for someone else to use. Remember the motto,

**Typing is no substitute for thinking.**

2.5 Errors and Debugging

It may occasionally happen that the first run of a new problem will be error-free and give the correct answers. But it is much more common that errors will be present and have to be corrected. Errors are of two types: Errors of form, or grammatical errors, that prevent even the running of the program; Logical errors in the program which cause wrong answers or even no answers to be printed.

Errors of form will cause error messages to be printed out instead of the expected answers. These messages give the nature of the error, and the line number in which the error occurred. Logical errors are often much harder to uncover, particularly when the program appears to give nearly correct answers. But after careful analysis and when the incorrect statement or statements are discovered, the correction is made by retyping the incorrect line or lines, by inserting new lines, or by deleting existing lines. These three kinds of corrections are made as follows:

- **Changing a line**
  - Type it correctly with the same line number.

- **Inserting a line**
  - Type it with a line number between those of the two existing lines.

- **Deleting a line**
  - Type the line number only.
Notice that being able to insert a line requires that the original line numbers not be consecutive numbers. For this reason, most users will start out using line numbers that are multiples of five or ten, but that is up to them.

These corrections can be made at any time, either before or after a run. They may even be made in an earlier part of the program while you are typing the later lines. Simply retype the offending line with its original line number, and then continue typing the rest of the program.

The whole process of locating errors or "debugging" a program is illustrated by a case history which starts on the next page. It takes us from the HELLOφ sequence to the final successful printing of the correct answers. The circled numbers refer to comments, which start below. For convenience, the portions typed by the computer are underlined or margin-lined, although no underlining is used on the actual computer.

The problem is to locate the maximum point on the sine curve between 0 and 3 by searching along the x-axis. The searching will be done three times, first with a spacing of 0.1, then with spacings of 0.01, and 0.001. In each case will be printed the location of the maximum, the maximum, and the spacing. The program as first written down on paper was:

```
  10  READ D
  20  LET X0 = 0
  30  FOR X = 0 TO 3 STEP D
  40  IF SIN(X) ≤ M THEN 100
  50  LET X0 = X
  60  LET M = SIN(X0)
  70  PRINT X0, X, D
  80  NEXT X0
  90  GO TO 20
 100  DATA .1, .01, .001
 110  END
```
HELI
USER NUMBER--999999
SYSTEM--BASIC
NEW OR OLD--NEW
NEW PROBLEM NAME--MAXSIN
READY.

10 READ D
20 LET X0 = 0
30 FOR X = 0 TO 3 STEP D
40 IF SIN(X) <= M THEN 100
50 LET X0 = X
60 LET M = SIN(X)
70 PRINT X0, X, D
80 NEXT X0
90 GOTO 20
20 LET X0 = 0
100 DATA .1, .01, .001
110 END
RUN

USER NO. 999999 PROBLEM NAME: MAXSIN 6 SEPT. 1964 TIME: 21:37

ILLEGAL FORMULA IN 70
NOT MATCHED WITH FOR IN 80
FOR WITHOUT NEXT

TIME: 0 SECS.

70 PRINT X0, X, D
40 IF SIN(X) <= M THEN 80
80 NEXT X
RUN

USER NO. 999999 PROBLEM NAME: MAXSIN 6 SEPT. 1964 TIME: 21:38
STOP
READY.
20 LET \( m = -1 \)
RUN

USER NO. 999999  PROBLEM NAME: MAXSIM 6 SEPT. 1964  TIME: 21:42

\[
\begin{array}{c|c|c}
0 & 0 & 1 \\
.1 & 1 & 1 \\
.2 & 2 & 1 \\
.3 & 3 & 1 \\
.4 & & \\
\end{array}
\]
STOP.
READY.

70 PRINT X0, M, D
RUN

USER NO. 999999  PROBLEM NAME: MAXSIM 6 SEPT. 1964  TIME: 21:43

\[
\begin{array}{c|c|c}
1.6 & 999574 & 1 \\
1.6 & 999574 & 1 \\
1.6 & 99 & \\
\end{array}
\]
STOP.
READY.

90 GO TO 10
5 PRINT "X VALUE", "SINE2-", RESOLUTION
RUN

USER NO. 999999  PROBLEM NAME: MAXSIM 6 SEPT. 1964  TIME: 21:44

INCORRECT FORMAT IN 5
TIME: 1 SECS.
5    PRINT "X VALUE", "SINE", "RESOLUTION"
RUN

USER Nº. 999999   PROBLEM NAME: MAXSIN   6 SEPT. 1964   TIME: 21:46

<table>
<thead>
<tr>
<th>X VALUE</th>
<th>SINE</th>
<th>RESOLUTION</th>
</tr>
</thead>
<tbody>
<tr>
<td>1.6</td>
<td>.999574</td>
<td>.1</td>
</tr>
<tr>
<td>1.57</td>
<td>1.</td>
<td>.01</td>
</tr>
<tr>
<td>1.571</td>
<td>1.</td>
<td>.001</td>
</tr>
</tbody>
</table>

TIME:  1 MINS.  0 SECS.

LIST

USER Nº. 999999   PROBLEM NAME: MAXSIN   6 SEPT. 1964   TIME: 21:48

5    PRINT "X VALUE", "SINE", "RESOLUTION"
10   READ D
20   LET M = -1
30   FOR X = 0 TO 3 STEP D
40   IF SIN(X) <= M THEN 80
50   LET X0 = X
60   LET M = SIN(X)
80   NEXT X
85   PRINT X0, M, D
90   G0 TO 10
100  DATA .1, .01, .001
110  END

SAVE

READY.
1. Notice the use of the backwards arrow to correct mistakes as you go along.

2. The user notices at this point that he had mistyped the word LET earlier, and corrects it.

3. An inspection of statement 70 shows that a variable XØ is used, which is illegal, when X0 was intended. The line is retyped correctly.

4. The variable in the NEXT statement should have been X instead of X0, and the change is made.

5. By chance, the user notices that originally the IF-THEN statement pointed to a DATA statement, and the correction is made.

6. The problem runs for a long time without any output. Since we expect output almost immediately, we suspect something is wrong. It must be that the PRINT statement is forever bypassed. This could happen only if M were so large that the IF-THEN statement was always satisfied. We then observe that we forgot to initialize M to some value less than the maximum value on the sine curve, so we choose -1.

7. At last we get printed output, but it appears that the printing is taking place each time through the loop rather than at the end of the loop. We move the print statement from before to after the NEXT statement, and incidentally change it to print M rather than X as the second term.

8. Ugh. Still not correct. We seem to be doing the first case over and over again. An infinite loop! This is corrected by going back to statement 10 instead of statement 20.

9. While we are at it, we put in labels to identify each column in the printed output.

10. Forgot the opening " for the third label.

11. We finally obtain the desired answers.

12. The final corrected version of the program is listed.

13. The program is saved for later use. (This should not be done unless future use is necessary.)

2.6 **Summary of Elementary BASIC Statements**

This section gives a short and concise but complete description of each of the nine types of BASIC statements discussed earlier in this chapter.
The notation \(< \ldots >\) is used to denote a particular unspecified instance of the type of thing referred to inside the \(< >\). Thus, \(<\text{line number}>\) is used to stand for any particular line number. \(<\text{variable}>\) refers to any variable, which is a single letter possibly followed by a single digit. \(<\text{expression}>\) stands for any particular expression, no matter how complicated, so long as it follows the rules for forming expressions given in section 2.2. \(<\text{number}>\) stands for any constant or data number.

2.6.1 LET

Form: \(<\text{line number}>\) LET \(<\text{variable}>\) = \(<\text{expression}>\)

Example: 100 LET X = X + 1
259 LET W7 = (W - X4↑3)*(Z - A1/(A - B)) - 17

Comment: The LET statement is not a statement of algebraic equality, but is rather a command to the computer to perform certain computations and to assign the answer to a certain variable. Thus, the first example tells the computer to take the current value of X, add 1 to it, and assign the answer to the variable X. In other words, X is increased by unity.

2.6.2 READ and DATA

Form: \(<\text{line number}>\) READ \(<\text{list of variables}>\)

Example: 150 READ X, Y, Z, X1, Y2, Z(K+I, J)

Form: \(<\text{line number}>\) DATA \(<\text{list of numbers}>\)

Example: 300 DATA 4, 2, 1.5, 0.6734E-2, -174, 321

Comment: A READ statement causes the variables listed in it to be given in order the next available numbers in the collection of DATA statements.

Comment: Before the program is run, the computer takes all the DATA statements in the order in which they appear and creates a large data block. Each time a READ statement is encountered anywhere in the program, the data block supplies the next available number or numbers. If the data block runs out of data, with a READ statement still asking for more, the program is assumed to be done.
2.6.3 PRINT

Form: <line number> PRINT <list of expressions to be printed>

Example: 100 PRINT X, Y, Z, B*B - 4*A*C, EXP(LOG(17))

Form: <line number> PRINT "<any string of characters>"

Example: 200 PRINT "THIS PROGRAM IS NOT GOOD."

150 PRINT "COMPUTES X + Y = Z"

Comment: The numerical quantities printed need not be simple variables, they may be any expressions. The expression is first evaluated, then printed. There may be any number of expressions separated by commas, but they will be printed five to a line.

Example: 150 PRINT "X", "Y", "Z"

Comment: Several messages may be included in the list separated by commas. The effect is to print the letter X in the first column, the letter Y in the 16th column, and the letter Z in the 31st column.

Example: 200 PRINT "X = ", X, "Y = ", Y

Comment: Labels and expressions may appear in the same print statement.

Comment: Much more variety is permitted in PRINT statements than is shown here. The additional flexibility is explained in section 3.1.

2.6.4 GO TO and IF-THEN

Form: <line number> GO TO <line number>

Example: 150 GO TO 75

240 GO TO 850

Comment: Sometimes called an unconditional go to, GO TO is used to interrupt the normal sequence of executing statements in the increasing order of their line numbers.

Form: <line number> IF <expression> <relation> THEN <line number>

Example: 140 IF X > Y + Z THEN 200

85 IF X * SIN(X) >= 1 THEN 100
Comment: Sometimes called a conditional go to, the IF-THEN statement provides a way to select one of two sequences in the program depending on the results of some previous computation. If the condition is met, the implied go to is performed; if the condition is not met, the next statement in sequence is performed.

Any of the six standard relations may be used.

<table>
<thead>
<tr>
<th>Symbol</th>
<th>Meaning</th>
</tr>
</thead>
<tbody>
<tr>
<td>&lt;</td>
<td>less than</td>
</tr>
<tr>
<td>&lt;=</td>
<td>less than or equal</td>
</tr>
<tr>
<td>=</td>
<td>equal</td>
</tr>
<tr>
<td>&gt;=</td>
<td>greater than or equal</td>
</tr>
<tr>
<td>&gt;</td>
<td>greater than</td>
</tr>
<tr>
<td>&lt;&gt;</td>
<td>not equal</td>
</tr>
</tbody>
</table>

2.6.5 FOR and NEXT

Form: `<line number>` FOR `<variable>` = `<expression>` TO `<expression>`

or

`<line number>` FOR `<variable>` = `<expression>` TO `<expression>` STEP `<expression>`

Example: 120 FOR X4 = (17 + COS(A))/3 TO 3*SQR(10) STEP 1/4

; (This represents the body of the loop.)

; 235 NEXT X4

Comment: Omitting the STEP part of the FOR statement is equivalent to having the stepsize equal to unity.

Comment: The above example will, assuming A to be equal to 0, cause the body of the loop to be performed several times, first with X4 equal to 6, next with X4 equal to 6.25, then 6.50, and so on. The last time the body of the loop will be performed is with X4 equal to 9.25, which is less than or equal to the final value 9.486 (approximately).

The FOR statement goes into the body of a loop if the variable has a value less than or equal to the final
value (in the case of a positive stepsize), or if the variable has a value greater than or equal to the final value (in the case of a negative stepsize.)

Upon leaving the loop, the program continues with the statement following the NEXT; the variable used in the FOR statement then has the value it had during the last passage through the loop (9.25 in the above example.)

Example: 240 FOR X = 8 TO 3 STEP -1

Comment: The body of the loop is performed with X equal to 8, 7, 6, 5, 4, and 3, and X has the value 3 upon leaving the loop.

Example: 456 FOR J = -3 TO 12 STEP 2

Comment: The body of the loop will be performed with J equal to -3, -1, 1, 3, 5, 7, 9, and 11. J will have the value 11 upon leaving the loop.

Example: 50 FOR Z = 2 TO -2

Comment: The body of the loop will not be performed. Instead, the computer will proceed to the statement immediately following the corresponding NEXT. The value of Z will then be 1, which is the initial value (2) minus the step size (1).

2.6.6 END

Form: \langle line number \rangle END

Example: 999 END

Comment: An END statement is required in all programs. It must also be the statement with the highest line number in the program.
3.1 More About PRINT

One of the conveniences of BASIC is that the format of answers is automatically supplied for the beginner. The PRINT statement does, however, permit a greater flexibility for the more advanced programmer who wishes to specify a more elaborate output.

The teletype line is divided into five zones of fifteen spaces each by BASIC, allowing the printing of up to five numbers per line. Three simple rules control the use of these zones.

1. A label, in quotes, is printed just as it appears.

2. A comma is a signal to move to the next print zone, or to the first print zone of the next line if it has just filled the fifth print zone.

3. The end of a PRINT statement signals a new line, unless a comma is the last symbol.

Each number occupies one zone. Each label occupies a whole number of zones; if it occupies part of a zone, the rest of the zone is filled with blanks. If a label runs through the fifth zone, part of it may be lost.

The examples on the following pages illustrate some of the various ways in which the PRINT statement can be used. It should be noted that a blank PRINT statement causes the typewriter to move to the next line, as is implied by rule 3 above.

The format in which BASIC prints numbers is not under the control of the user. However, the following rules may be used to guide the program-
NEW
NEW PROBLEM NAME--PRINTER
READY.

10 RA DELETED
10 READ A, B
20 PRINT "FIRST NO. ="A, "SECOND NO. ="B
30 DATA 2.3, -3.17
40 END
RUN

USER NO. 999999 PROBLEM NAME: PRINTE 6 SEPT. 1964 TIME: 21:52
FIRST NO. = 2.3 SECOND NO. = -3.17
TIME: 0 SECS.

Comment: Notice that with no comma between the label and the variable name, the label and the value of the variable appear together. But since the label and the number (with its unprinted non-significant zeros) occupies more than one zone, the second answer is printed starting in the third zone. This is in accordance with rule 2, which says that we do not start a new zone until a comma is encountered.

SCRATCH
READY.

10 FOR I = 1 TO 12
20 PRINT I,
30 NEXT I
40 END
RUN

USER NO. 999999 PROBLEM NAME: PRINTE 6 SEPT. 1964 TIME: 21:53
1 2 3 4 5
6 7 8 9 10
11 12
TIME: 0 SECS.
5 PRINT "THIS PROGRAM COMPUTES AND PRINTS THE NTH POWERS"
6 PRINT "OF THE NUMBERS LESS THAN OR EQUAL TO N FOR VARIOUS"
7 PRINT "N FROM 1 THROUGH 7."
8 PRINT
10 FOR N = 1 TO 7
15 PRINT "N = "N
20 FOR I = 1 TO N
30 PRINT I**N,
40 NEXT I
50 PRINT
60 PRINT
70 NEXT N
80 END

RUN

USER NO. 999999 PROBLEM NAME: PRINT 6 SEPT. 1964 TIME: 22:09

THIS PROGRAM COMPUTES AND PRINTS THE NTH POWERS
OF THE NUMBERS LESS THAN OR EQUAL TO N FOR VARIOUS
N FROM 1 THROUGH 7.

N = 1
1

N = 2
1 4.

N = 3
1 1 8.

N = 4
1 1 16. 27.

N = 5
1 1 1 32. 243. 1024 3125.

N = 6
1 1 64. 729. 4096 15625.
1 46656.

N = 7
1 1 128. 2187. 16384 78125.
1 279936. 823543.
TIME: 2 SECS.
mer in interpreting his printed results.

1. No more than six significant digits are printed (except for integers -- see rule 4.)

2. Any trailing zeros after the decimal point are not printed.

3. For numbers less than 0.1, the form X. Xxxxx E-Y is used unless the entire significant part of the number can be printed as a six decimal number. Thus, .03456 means that the number is exactly .0345600000, while 3.45600 E-2 means that the number has been rounded to .0345600.

4. If the number is an exact integer, the decimal point is not printed. Furthermore, integers of up through nine digits are printed in full.

A packed form of output is available by using the character ";'" instead of ";". Briefly, whereas ";" tells the computer to move to the next zone for the next answer, ; tells the computer to move to the beginning of the next multiple of three characters for the next answer instead of to the next zone, with six characters being the minimum. One can thus pack many more than five numbers on a line if the numbers themselves require less than a full zone to print. An example of this option is shown on the next page.

With packed output using the semi-colon, the programmer can print 11 three digit numbers per line, 8 six digit numbers per line, or 6 nine digit numbers per line. Mixtures of the three types in subsequent lines may not line up, as the example shows. The user should be careful about using the semi-colon with full length numbers which might occur near the end of a print line. BASIC checks to see if there are 12 or more spaces at the end of a line before printing a number there, but some numbers require 15 spaces. The same warning holds for printing labels near the end of the line. In each case, the last few characters may be lost.
```
10 FOR I = 1 TO 100
20 PRINT I*I*I;
30 NEXT I
40 END
RUN

USER NO. 999999    PROBLEM NAME: PRINT  6 SEPT. 1964    TIME: 22:14

<table>
<thead>
<tr>
<th>1</th>
<th>8</th>
<th>27</th>
<th>64</th>
<th>125</th>
<th>216</th>
<th>343</th>
<th>512</th>
<th>729</th>
<th>1000</th>
<th>1331</th>
</tr>
</thead>
<tbody>
<tr>
<td>1728</td>
<td>2197</td>
<td>2744</td>
<td>3375</td>
<td>4096</td>
<td>4913</td>
<td>5832</td>
<td>6859</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>8000</td>
<td>9261</td>
<td>10648</td>
<td>12167</td>
<td>13824</td>
<td>15625</td>
<td>17576</td>
<td>19683</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>21952</td>
<td>24389</td>
<td>27000</td>
<td>29791</td>
<td>32768</td>
<td>35937</td>
<td>39304</td>
<td>42875</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>46656</td>
<td>50653</td>
<td>54872</td>
<td>59319</td>
<td>64000</td>
<td>68921</td>
<td>74088</td>
<td>79507</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>85184</td>
<td>91125</td>
<td>97336</td>
<td>103823</td>
<td>110592</td>
<td>117649</td>
<td>125000</td>
<td>132651</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>140608</td>
<td>148877</td>
<td>157464</td>
<td>166375</td>
<td>175616</td>
<td>185193</td>
<td>195112</td>
<td>205379</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>216000</td>
<td>226981</td>
<td>238328</td>
<td>250047</td>
<td>262144</td>
<td>274625</td>
<td>287496</td>
<td>300763</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>314432</td>
<td>328509</td>
<td>343000</td>
<td>357911</td>
<td>373248</td>
<td>389017</td>
<td>405224</td>
<td>421875</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>438976</td>
<td>456533</td>
<td>474552</td>
<td>493039</td>
<td>512000</td>
<td>531441</td>
<td>551368</td>
<td>571787</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>592704</td>
<td>614125</td>
<td>636056</td>
<td>658503</td>
<td>681472</td>
<td>704969</td>
<td>729000</td>
<td>753571</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>778688</td>
<td>804357</td>
<td>830584</td>
<td>857375</td>
<td>884736</td>
<td>912673</td>
<td>941192</td>
<td>970299</td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

TIME:    2 SECS.
```
3.2 Lists and Tables

In addition to the ordinary variables used by BASIC, there are variables that can be used to designate lists or tables. For instance, \( A(7) \) would denote the seventh item in a list called \( A \); \( B(3,7) \) denotes the item in the third row and seventh column of the table called \( B \). We commonly write \( A_7 \) and \( B_{3,7} \) for those same items, and use the term subscripts to denote the numbers that point to the desired items in the list or table. (The reader may recognize that lists and tables are called, respectively, vectors and matrices by mathematicians.)

The name of a list or table must be a single letter. The subscripts may be any expression, no matter how complicated, as long as they have non-negative integer values. The single letter denoting a list or table name may also be used to denote a simple variable without confusion. However, the same letter may not be used to denote both a list and a table in the same program. The following are acceptable examples of list and table items, though not necessarily in the same program:

\[
B( I + K ) \quad B( I, K ) \quad Q( A(3, 7), B - C )
\]

The example on the next page shows a simple use of both lists and tables in the same program. We might think of this program as one that computes the total sales for each of five salesmen selling three different goods. The list \( P \) gives the price of the three goods. The table \( S \) gives the individual item sales of the five salesmen, where the rows stand for the items and the columns for the salesmen. We assume that the items sell for \$1.25, 4.30, and 2.50, respectively, and that salesman 1 sold 40 of item 1, 10 of item 2, and 35 of item 3, and so on.
10 FOR I = 1 TO 3
20 READ P(I)
30 NEXT I
40 FOR I = 1 TO 3
50 FOR J = 1 TO 5
60 READ S(I,J)
70 NEXT J
80 NEXT I
90 FOR J = 1 TO 5
100 LET S = 0
110 FOR I = 1 TO 3
120 LET S = S + P(I) * S(I,J)
130 NEXT I
140 PRINT "TOTAL SALES FOR SALESMAN " , "$" , S
150 NEXT J
200 DATA 1.25, 4.30, 2.50
210 DATA 40, 20, 37, 29, 42
220 DATA 10, 16, 3, 21, 8
230 DATA 35, 47, 29, 16, 33
300 END

RUN

TOTAL SALES FOR SALESMAN 1 $ 180.5
TOTAL SALES FOR SALESMAN 2 $ 211.3
TOTAL SALES FOR SALESMAN 3 $ 131.65
TOTAL SALES FOR SALESMAN 4 $ 166.55
TOTAL SALES FOR SALESMAN 5 $ 169.4

TIME: 1 SECS.
By way of explanation, lines 10 through 30 read in the values of the list P. Lines 40 through 80 read in the values of the table S.

Lines 90 through 150 compute the total sales for the five salesmen and print each answer as it is computed. The computation for a single salesman takes place in lines 100 through 130. In lines 90 through 150, the letter I stands for the good number, and the letter J stands for the salesman number.

BASIC provides that each list has a subscript running from 0 to 10, inclusive. Each subscript in a table may run from 0 to 10. If the user desires to have larger lists or tables, he may use a DIM statement in his program. For example,

10  DIM A(17)

indicates to the computer that the subscript of the list A runs from 0 to 17, inclusive; similarly,

20  DIM B(15, 20), S(3)

means that the subscripts of B run from 0 through 15 for rows, and 0 through 20 for columns, and that the subscript of the list S runs from 0 through 3. The numbers used to denote the size of a list or table in a DIM statement must be integer numbers. The DIM statement is used not only to indicate that lists and tables are larger than 0-10 in each subscript, but also to allocate storage space in very large programs by telling the computer that only, say, 4 spaces are needed for the list S as shown above.

It should be mentioned that using a DIM statement does not require the user to use all of the spaces so allocated.
3.3 Functions and Subroutines

Two additional functions that are in the BASIC repertory but which were not described in section 2.2 are INT and RND. INT is used to determine the integer part of a number that might not be a whole number. Thus INT(7.8) is equal to 7. As with the other functions, the argument of INT may be any expression. One use of INT is to round numbers to the nearest whole integer. If the number is positive, use INT(X + .5). The reader should verify that this process is equivalent to the familiar process of rounding. If the number is negative, INT(X - .5) must be used. The reason is that INT(-7.8) is -7, not -8. INT always operates by chopping off the fractional part, whether the number is positive or negative.

INT can be used to round to any specific number of decimal places. Again, for positive numbers,

\[ \text{INT}(100\times X + .5)/100 \]

will round X to the nearest correct two decimal number.

The function RND produces a random number between 0 and 1. The form of RND requires an argument, though the argument has no significance; thus, we commonly choose a single letter such as X or Z, and use RND(X).

The property of RND is that it produces a new and different random number each time it is used in a program. Thus, to produce and print 20 random digits, one might write a program like that shown on the next page.

The middle example shows that the next time the program is run, the same sequence is obtained. To vary the sequence we might "throw away" an arbitrary number of random numbers at the start of the program. In the third example at the bottom of the page, the first 115 random numbers are
10 FOR I = 1 TO 20
20 PRINT INT(10*RND(X));
30 NEXT I
40 END
RUN

USER NO. 999999 PROBLEM NAME: RANDOM 6 SEPT. 1964 TIME: 23:33
9 0 4 1 6 4 9 5 8 3 8
6 6 9 3 4 0 7 7 7
TIME: 1 SECS.

RUN

USER NO. 999999 PROBLEM NAME: RANDOM 6 SEPT. 1964 TIME: 23:34
9 0 4 1 6 4 9 5 8 3 8
6 6 9 3 4 0 7 7 7
TIME: 1 SECS.

10 READ N
20 FOR I = 1 TO N
30 LET X = RND(X)
40 NEXT I
50 FOR I = 1 TO 20
60 PRINT INT(10*RND(X));
70 NEXT I
80 DATA 115
90 END
RUN

USER NO. 999999 PROBLEM NAME: RANDOM 6 SEPT. 1964 TIME: 23:35
7 9 9 0 9 4 5 9 9 7 5
2 2 4 2 0 0 0 3 3
TIME: 1 SECS.
discarded. The output shows random digits numbered 116 through 135.

Additional flexibility is provided in BASIC by three statements that permit the use of user-defined functions and subroutines.

The DEF statement permits the user to define a function other than the standard functions listed in section 2.2 so that he doesn't have to keep repeating the formula for the function each time he uses it in his program. The name of a defined function must be three letters, the first two of which are FN. The user thus may define up to 26 functions. The following examples illustrate the form of the DEF statement:

25 DEF FNF(Z) = SIN(Z*P)  (where P has the value 3.14159265/180)
40 DEF FNL(X) = L0G(X)/L0G(10)

Thus, FNF is the sine function measured in degrees, and FNL is the function log-to-the-base-ten.

The DEF statement may occur anywhere in the program. The user needs to be cautioned that the variable used in the DEF statement must not be subscripted, and that it is used every time that function is used. Thus, in a program containing FNF as above defined, it is best not to use the variable Z elsewhere in the program.

The expression on the right of the equal sign can be any expression that can be fit into one line. It could involve many other variables besides the one denoting the argument of the function. Thus,

60 DEF FNX(X) = SQR(X*X + Y*Y)

may be used to set up a function that computes the square root of the sum of the squares of X and Y. To use FNX, one might use the following:

10 LET Y = 30
20 LET S1 = FNX(40)
Of course, S1 would end up having the value 50.

It should be noted that one does not need DEF unless the defined function must appear at two or more locations in the program. Thus,

```
10  DEF FNF(Z) = SIN(Z*P)
20  LET P = 3.14159265/180
30  FOR X = 0 TO 90
40  PRINT X, FNF(X)
50  NEXT X
60  END
```

might be more efficiently written as

```
20  LET P = 3.14159265/180
30  FOR X = 0 TO 90
40  PRINT X, SIN(X*P)
50  NEXT X
60  END
```

to compute a table of values of the sine function in degrees.

The use of DEF is limited to those cases where the value of the function can be computed within a single BASIC statement. Often much more complicated functions, or perhaps even pieces of program that are not functions, must be calculated at several different points within the program. For this, the GOSUB statement may frequently be useful.

The form of a GOSUB statement is illustrated as follows:

```
25  GOSUB 180
```

The effect of the GOSUB is exactly the same as a GOTO except that note is taken by the computer as to where the GOSUB statement is in the program. As soon as a RETURN statement is encountered, the computer automatically goes back to the statement immediately following the GOSUB. As a skeleton example,

```
100  LET X = 3.
110  GOSUB 400
120  PRINT U, V, W
```

(continued on next page)
200 LET X = 5
210 GOSUB 400
220 LET Z = U + 2*V + 3*W

400 LET U = X*X
410 LET V = X*X*X
420 LET W = X*X*X*X + X*X*X + X*X + X
430 RETURN

When statement 400 is entered by the GOSUB 400 in line 110, the computations in lines 400, 410, and 420 are performed, after which the computer goes back to statement 120. When the subroutine is entered from statement 210, the computer goes back to statement 220.

The user must be very careful not to write a program in which a GOSUB appears inside a subroutine which itself is entered via a GOSUB; it just won't work.

As a complete illustration, the next page contains a program that determines the Greatest Common Divisor of three integers, using the celebrated Euclidean algorithm as a subroutine. The subroutine is contained in lines 200 to 310, and is applied to two integers only. The main routine applies this subroutine to the first two integers, and then to the GCD of these and the third integer. The GCD is then printed, and a new case considered.

3.4 Some Ideas for More Advanced Programmers

An important part of any computer program is the description of what it does, and what data should be supplied. This description is commonly called documentation. One of the ways a computer program can be documented is by supplying remarks along with the program itself. BASIC provides for this capability with the REM statement. For example,
10 PRINT "A", "B", "C", "GCD"
20 READ A, B, C
30 LET X = A
40 LET Y = B
50 GOSUB 200
60 LET X = G
70 LET Y = C
80 GOSUB 200
90 PRINT A, B, C, G
100 GO TO 20
110 DATA 60, 90, 120
120 DATA 38456, 64872, 98765
130 DATA 32, 384, 72
200 LET Q = INT(X/Y)
210 LET R = X - Q*Y
220 IF R = 0 THEN 300
230 LET X = Y
240 LET Y = R
250 GO TO 200
300 LET G = Y
310 RETURN
999 END

RUN

USER NO. 999999     PROBLEM NAME: GCD3NO     6 SEPT. 1964     TIME: 23:28

<table>
<thead>
<tr>
<th>A</th>
<th>B</th>
<th>C</th>
<th>GCD</th>
</tr>
</thead>
<tbody>
<tr>
<td>60</td>
<td>90</td>
<td>120</td>
<td>30</td>
</tr>
<tr>
<td>38456</td>
<td>64872</td>
<td>98765</td>
<td>1</td>
</tr>
<tr>
<td>32</td>
<td>384</td>
<td>72</td>
<td>8.</td>
</tr>
</tbody>
</table>

TIME: 1 SECS.
rem this program solves linear equations of the form
rem \( A_1 x_1 + A_2 x_2 = B_1 \), \( A_3 x_1 + A_4 x_3 = B_2 \). the data
rem must first list the four values of \( a \) in order, then
rem the desired right hand sides for which solutions
rem are needed.

might reasonably be added to the original example for solving linear equa-
tions. For longer programs, more detailed REM's may be needed, espe-
sially ones spotted throughout the program to remind you what each of the
parts does. Each user quickly learns how much documentation he needs to
permit him to understand his program, and where to put REM statements.
But it is certain that REM's are needed in any saved program. It should be
emphasized that REM's have absolutely no effect on the computation.

Sometimes a program will have two or more natural ending points.
In such a case the programmer might use a G\( \phi \)\( \phi \) to the END statement.
Such a statement can be replaced by a STOP, with nothing following the
word STOP. Thus,

\begin{verbatim}
400  G\( \phi \)\( \phi \)  999
710  G\( \phi \)\( \phi \)  999
999   END
\end{verbatim}

may be replaced by

\begin{verbatim}
400   STOP
710   STOP
999   END
\end{verbatim}

BASIC allows G\( \phi \) T\( \phi \) and IF-THEN statements to point to REM and
DATA statements. The effect is to perform a vacuous statement having
that number and proceed to the next numbered statement. In the case of DATA statements, the END statement might eventually be reached. However, for REM statements the programmer might deliberately have his GOTO's point to REM statements, the remark part identifying that part of the program.

One of the most important and difficult problems in computing is that of round-off error. It exerts its influence in subtle ways, and sometimes in ways not so subtle. A full treatment of the effects of round-off error is beyond the scope of this manual, but one fairly common situation will be discussed.

Most programmers eventually write or encounter a program something like this:

```
  5  LET S = 0
  10  LET X = 0
  20  LET S = S + X
  30  IF X >= 2 THEN 60
  40  LET X = X + .1
  50  GOTO 20
  60  PRINT S
  70  END
```

for computing the sum of all the non-negative multiples of .1 less than or equal to 2. The correct answer is 21, but invariably the program will produce 23.1 as the answer. What is wrong? Round-off has reared its ugly head high enough for us to see. The explanation is that the computer works in the binary number system, and cannot express .1 exactly. Just as 1/3 cannot be expressed in terms of a single decimal number, neither can .1 be expressed in terms of a single binary number. It turns out that .1 in the computer is a number very slightly less than .1. Thus, when the loop in the above example has been performed 21 times, the value of X is not 2 exactly, but is very slightly less than 2. The IF statement in line 30
determines that the final value, exactly 2, has not yet been achieved or exceeded, and so calls for one more passage through the loop.

If the programmer had known that the computer treats 1 as a number slightly less, he could have compensated by writing 1.95 in place of 2 in statement 30. A better way rests on the fact that the computer performs exactly correct arithmetic for integers. The user may thus count the number of times through the loop with integers. The example may be rewritten as follows:

```
5  LET S = 0
10  LET N = 0
20  LET S = S + N/10
30  IF N >= 20 THEN 60
40  LET N = N + 1
50  GOTO 20
60  PRINT S
70  END
```

Better still, a FOR statement can shorten the program to

```
10  LET X = 0
20  FOR N = 1 TO 20
30  LET S = S + N/10
40  NEXT N
50  PRINT S
60  END
```

One of the most exasperating problems confronting programmers is that of a fairly long and complex program that looks as if it should work simply refuses to do so. (Presumably, all errors of form have been detected and removed.) The locating and removing of logical errors is called debugging, and the methods to be used depend on the nature of the program and also on the programmer himself. An important part of debugging is intuition, but it is possible to suggest some approaches that might be useful in many cases.

The first thing to do with an apparently incorrect program is
to check very carefully the method used. If that doesn't uncover the bug, then examine very carefully your programming to see if you have mixed up any of the variables. It is often difficult to spot such errors because one tends to see in a program what he expects to see rather than what is there.

Another method that is extremely useful in providing clues as to the nature and location of the bug or bugs is tracing. In BASIC this tracing may be accomplished by inserting superfluous PRINT statements at various places in your program to print the values of some of the intermediate quantities. When the program is then RUN, the values of these intermediate quantities often suggest the exact nature of the error. When the program has been debugged and is working properly, these statements are removed.

There are some matters that do not affect the correct running of programs, but pertain to style and neatness. For instance, as between two or more similar ways to prepare a part of a program, one should select the one that is most easily understood unless there is an important reason not to do so.

More experienced programmers will tend to group the data in DATA statements so that it reflects the READ statements that correspond. The first example on linear equations represents bad style, but was done purposely to illustrate that one can arbitrarily group the data in the DATA statements.

One tends after a while to place his data statements near the end of the program, or near the beginning, but at least in one group to avoid confusing himself with DATA statements spread throughout the program.
Some programmers also tend to give the END statement a number like 9999 to insure that it will be the one with the highest number.

No doubt the user will be able to devise other ways to make a program neat and readable. But again, the important consideration in style is to program in a way that makes it more understandable and useful to both oneself and others in the future.
4.1 **Purpose**

A card-operated (on-line) version of BASIC is available and provides the following advantages over teletypes:

1. Longer programs are allowed.
2. There is no limit on DATA.
3. Output is on the highspeed printer. This is much faster and allows 8 numbers per line in the normal format, and up to 18 in packed format.
4. Matrix subroutines are available.

4.2 **How to Prepare a Deck.**

You punch on cards a program almost exactly the way you would type it on a teletype, with one instruction per card. Due to the fact that there are fewer symbols on a key-punch, the following modifications are needed:

In PRINT, use single quote (') in place of quote (").

For the relational symbols, use the following three letter equivalents.

<table>
<thead>
<tr>
<th>Symbol</th>
<th>Equivalent</th>
</tr>
</thead>
<tbody>
<tr>
<td>=</td>
<td>EQU</td>
</tr>
<tr>
<td>&lt;</td>
<td>LSS</td>
</tr>
<tr>
<td>&gt;=</td>
<td>GR T</td>
</tr>
<tr>
<td>&lt;=</td>
<td>LQU</td>
</tr>
<tr>
<td>&gt;</td>
<td>GQU</td>
</tr>
<tr>
<td>&lt;&gt;</td>
<td>NQU</td>
</tr>
</tbody>
</table>

In place of a semi-colon in a PRINT statement, you must use a 5-8 multiple punch.

4.3 **Differences in Operation.**

Do not type DATA as part of your program deck. Instead, add to the
end of your program (after the END card) a data deck. There are no line
numbers for data, and the word DATA should not appear. Just a list of
numbers, separated by commas. You may put as many or as few numbers
on one card as you wish.

There is a new instruction PAGE. This starts a new printed page
for the output.

There is a series of matrix subroutines available as part of the
new instruction MAT. See the next section for a detailed description.

You have much larger matrices available. The limitations are:

(1) At most 1000 components in any one vector or list.

(2) No matrix or table dimension may exceed 500.

(3) The total number of components in all vectors and
matrices (lists and tables) may not exceed 4000.

Matrices must start with component 1 (not 0).

4.4 MAT

Matrix subroutines are available in CARDBASIC. They are called
by means of the MAT instruction. Observe the following restrictions: (1)
A matrix in an MAT instruction must have had a DIM declared. (2) While
the same matrix may appear in an MAT and an arithmetic statement, it
must occur for the first time in an MAT statement.

Allowed MAT operations are:

<table>
<thead>
<tr>
<th>MAT</th>
<th>READ A(M, N)</th>
<th>Read one matrix, dimensions shown.</th>
</tr>
</thead>
<tbody>
<tr>
<td>MAT</td>
<td>PRINT A</td>
<td>Print one matrix.</td>
</tr>
<tr>
<td>MAT</td>
<td>C = A + B</td>
<td>Add two matrices.</td>
</tr>
<tr>
<td>MAT</td>
<td>C = A - B</td>
<td>Subtract matrices.</td>
</tr>
</tbody>
</table>

(continued on next page)
MAT  C = A * B    Multiply matrices.
MAT  C = ZER(M, N)  Introduce a 0 matrix, dimensions shown.
MAT  C = CN(M, N) Matrix of all 1's, dimensions shown.
MAT  C = IDN(N)   Identity matrix, dimension shown.
MAT  C = TRN(A)   Transpose.
MAT  C = INV(A)   Inverse.
MAT  C = (k)*A   Constant multiple, note parentheses.

In appropriate places, vectors may be substituted for matrices. E.g.,
MAT READ A(7) will read a 7-component column vector. MAT PRINT
prints all vectors as row-vectors, for convenience. M and N may be form-
ulas. Thus one has the possibility of writing

    DIM     A(20, 20)
    READ    M
    MAT     READ A(M, M)

This allows A to be any square matrix up to 20 x 20, its actual dimension
specified in the data. If MAT PRINT A is followed by a 5-8 multiple punch,
matrices will be printed in the ";;" packed format.
APPENDICES

APPENDIX A — Error Messages

The various error messages that can occur in BASIC, together with their interpretation, are now given:

<table>
<thead>
<tr>
<th>Error Message</th>
<th>Interpretation</th>
</tr>
</thead>
<tbody>
<tr>
<td>DIMENSION TØ Ø LARGE</td>
<td>The size of a list or table is too large for the available storage. Make them smaller. (See Appendix B.)</td>
</tr>
<tr>
<td>ILLEGAL CONSTANTE</td>
<td>More than nine digits or incorrect form in a constant number.</td>
</tr>
<tr>
<td>ILLEGAL FORMULA</td>
<td>Perhaps the most common error message, may indicate missing parentheses, illegal variable names, missing multiply signs, illegal numbers, or many other errors. Check the statement thoroughly.</td>
</tr>
<tr>
<td>ILLEGAL RELATION</td>
<td>Something is wrong with the relational expression in an IF-THEN statement. Check to see if you used one of the six permissible relational symbols.</td>
</tr>
<tr>
<td>ILLEGAL LINE NUMBER</td>
<td>Line number is of incorrect form, or contains more than five digits.</td>
</tr>
<tr>
<td>ILLEGAL INSTRUCTION</td>
<td>Other than one of the fifteen legal BASIC instructions has been used following the line number.</td>
</tr>
<tr>
<td>ILLEGAL VARIABLE</td>
<td>An illegal variable name has been used.</td>
</tr>
<tr>
<td>INCORRECT FORMAT</td>
<td>The format of an instruction is wrong. See especially IF-THEN's and FOR's.</td>
</tr>
<tr>
<td>END IS NOT LAST</td>
<td>Self-explanatory, it also occurs if there are two or more END statements in the program.</td>
</tr>
<tr>
<td>NO END INSTRUCTION</td>
<td>The program has no END statement.</td>
</tr>
<tr>
<td>Error Code</td>
<td>Description</td>
</tr>
<tr>
<td>--------------------------</td>
<td>-----------------------------------------------------------------------------</td>
</tr>
<tr>
<td>NO DATA</td>
<td>There is at least one READ statement in the program, but no DATA statements.</td>
</tr>
<tr>
<td>UNDEFINED FUNCTION</td>
<td>A function such as FNF ( ) has been used without appearing in a DEF statement. Check for typographical errors.</td>
</tr>
<tr>
<td>UNDEFINED NUMBER</td>
<td>The statement number appearing in a GO TØ or IF-THEN statement does not appear as a line number in the program.</td>
</tr>
<tr>
<td>PROGRAM TOO LONG</td>
<td>Either the program itself is too long for the available storage, or there are too many constants and printed labels. (See Appendix B.)</td>
</tr>
<tr>
<td>TOO MUCH DATA</td>
<td>There is too much data in the program. (See Appendix B.)</td>
</tr>
<tr>
<td>TOO MANY LABELS</td>
<td>The total length of all printed labels in the program exceeds the limit. (See Appendix B.)</td>
</tr>
<tr>
<td>TOO MANY LOOPS</td>
<td>There are too many FØR-NEXT combinations in the program. The upper limit is 26. (See Appendix B.)</td>
</tr>
<tr>
<td>NOT MATCH WITH FOR</td>
<td>An incorrect NEXT statement, perhaps with a wrong variable given. Also, check for incorrectly nested FØR statements.</td>
</tr>
<tr>
<td>FOR WITHOUT NEXT</td>
<td>A missing NEXT statement. This message can also occur in conjunction with the previous one.</td>
</tr>
<tr>
<td>CUT PROGRAM OR DIMS.</td>
<td>Either the program is too long, or the amount of space reserved by the DIM statements is too much, or a combination of these. This message can be eliminated by either cutting the length of the program, or by reducing the size of the lists and tables.</td>
</tr>
<tr>
<td>SUBSCRIPT ERROR</td>
<td>A subscript has been called for that lies outside the range specified in the DIM statement, or if no DIM statement applies, outside the range 0 through 10.</td>
</tr>
<tr>
<td>ILLEGAL RETURN</td>
<td>Occurs if a RETURN is encountered before the first GØSUB during the running of a program. (Note: BASIC does not require the GØSUB to have an earlier statement number -- only to perform a GØSUB before performing a RETURN.)</td>
</tr>
</tbody>
</table>
APPENDIX B -- Limitations on BASIC

There are some limitations imposed on BASIC by the limited amount of computer storage. Listed below are some of these limitations, in particular, those that are related to the error messages in APPENDIX A. The reader should realize that while the BASIC language itself is fixed, in time some of these limitations may be relaxed slightly.

<table>
<thead>
<tr>
<th>Item</th>
<th>Limitation</th>
</tr>
</thead>
<tbody>
<tr>
<td>Length of program</td>
<td>Difficult to relate to the BASIC program, but in general about two feet of teletype paper filled with BASIC statements is about it.</td>
</tr>
<tr>
<td>Constants and printed labels</td>
<td>The total number of constants and different printed labels must not exceed 175.</td>
</tr>
<tr>
<td>Data</td>
<td>There can be no more than 300 data numbers.</td>
</tr>
<tr>
<td>Length of printed labels</td>
<td>The total length of printed labels cannot exceed something slightly less than 600 characters.</td>
</tr>
<tr>
<td>FØR statements</td>
<td>There can be no more than 26 FØR statements in a program.</td>
</tr>
<tr>
<td>GØ TØ and IF-THEN statements</td>
<td>The total number of these statements combined cannot exceed 80.</td>
</tr>
<tr>
<td>Lists and Tables</td>
<td>The total number of elements in all the lists and tables combined cannot exceed 1500.</td>
</tr>
</tbody>
</table>
APPENDIX C -- Summary of the 15 BASIC Statements

In this summary it is assumed that all statements begin with a line number. Following each is one example.

LET

\[ \text{LET } \langle \text{variable} \rangle = \langle \text{expression} \rangle \]
\[ 10 \ \text{LET} \ X1 = Y + Z + (Z / A - B \uparrow D1) \]

READ

\[ \text{READ } \langle \text{variable} \rangle , \langle \text{variable} \rangle , \ldots , \langle \text{variable} \rangle \]
\[ 10 \ \text{READ} \ X, Y, Z , A1, Q(I, J) \]

DATA

\[ \text{DATA } \langle \text{number} \rangle , \langle \text{number} \rangle , \ldots , \langle \text{number} \rangle \]
\[ 10 \ \text{DATA} \ 1, 2, -3, 7, 123.479, -2.35E-4 \]

PRINT

\[ \text{PRINT } \langle \text{label} \rangle , \text{or } \langle \text{label} \rangle \langle \text{expression} \rangle , \text{or } \]
\[ 10 \ \text{PRINT} \ "\text{SINE}" , "X = " X(I, K), A + B*COS(Y) \]

GOTO

\[ \text{GOTO } \langle \text{line number} \rangle \]
\[ 10 \ \text{GOTO} \ 17 \]

IF-THEN

\[ \text{IF } \langle \text{expression} \rangle \langle \text{relational} \rangle \langle \text{expression} \rangle \text{ THEN } \langle \text{line number} \rangle \]
\[ 10 \ \text{IF} \ X + Y > 0 \ \text{THEN} \ 419 \]

FOR

\[ \text{FOR } \langle \text{unsubscripted variable} \rangle = \langle \text{expression} \rangle \text{ TO } \langle \text{expression} \rangle \text{ STEP } \langle \text{expression} \rangle \]
\[ 10 \ \text{FOR} \ I = 1 \text{ TO } 17 \]
\[ 10 \ \text{FOR} \ X1 = 0 \text{ TO } 7 \text{ STEP } 0.5 \]

NEXT

\[ \text{NEXT } \langle \text{unsubscripted variable} \rangle \]
\[ 10 \ \text{NEXT} \ X1 \]

(continued on next page)
(continued)

END

10 END

STOP

10 STOP

DEF

DEF FN 〈letter〉 〈unsubscripted variable〉) = 〈expression〉

10 DEF FNG(Z) = 1 + SQR(1 + Z * Z)

GOSUB

GOSUB 〈line number〉

10 GOSUB 110

RETURN

RETURN

10 RETURN

DIM

DIM 〈letter〉 (〈integer〉), or 〈letter〉 (〈integer〉, 〈integer〉)

10 DIM A(17), B(3, 20)

REM

REM 〈any string of characters whatsoever〉

10 REM THIS IS THE END OF APPENDIX C